Function Definition, Debugging, Memoization and Verification

Warren A. Hunt, Jr.
hunt@cs.utexas.edu

Computer Science Department
University of Texas
2317 Speedway, M/S D9500
Austin, TX 78712-0233

January, 2025

Function Definition and Memoization

Users of ACL2 often wish to confirm that some efficient, but complex implementation, of some computing process is equivalent to a simple, clear specification.

In this talk, we exhibit several ACL2 features that demonstrate functional equivalence between functions with very different efficiencies.

We will demonstrate ACL2's debugging facility.

We will demonstrate ACL2's memoization facility.

We will demonstrate ACL2's verification facility.

We will discuss how the techniques presented can aid software development.

Fibonacci Definition

Using ACL2, we can define the Fibonacci function.

```
(defun fib (x)
  (declare (xargs :guard (natp x)))
  (if (zp x)
      0
    (if (= x 1)
        1
      (+ (fib (- x 2))
         (fib (- x 1))))))
```

For this definition to be accepted, two termination conjectures must be checked – one for each inferior call to `fib`.

ACL2 can process this definition automatically, and observes that `fib` always returns a natural number.

Fibonacci Execution

The newly defined function, `fib`, can be executed immediately.

```
ACL2 !>(fib 10)
55
ACL2 !>(fib 20)
6765
```

However, when we evaluate the Fibonacci function with larger arguments, we must wait for the answer...

```
ACL2 !>(time$ (fib 50))
; (EV-REC *RETURN-LAST-ARG3* ...) took
; 191.77 seconds realtime, 188.34 seconds runtime
; (16 bytes allocated).
12586269025
```

ACL2 Function Tracing

We can investigate how our FIB function operates by tracing.

```
ACL2 !>(trace$ fib)
((FIB))
ACL2 !>(fib 3)
1> (ACL2_*1*_ACL2::FIB 3)
  2> (FIB 3)
    3> (FIB 1)
    <3 (FIB 1)
    3> (FIB 2)
      4> (FIB 0)
      <4 (FIB 0)
      4> (FIB 1)
      <4 (FIB 1)
    <3 (FIB 1)
  <2 (FIB 2)
<1 (ACL2_*1*_ACL2::FIB 2)
2
```

There are many options for tracing function execution. This facility can be used for debugging.

Alternative Fibonacci Definition

ACL2 permits an alternative function for execution.

```
(defun fib (x)
  (declare (xargs :guard (natp x)))
  (mbe
   :logic
   (if (zp x)
       0
     (if (= x 1)
         1
       (+ (fib (- x 2))
          (fib (- x 1)))))
   :exec
   (if (< x 10)
       (case x
         (0 0)
         (1 1)
         (2 1)
         (3 2)
         (4 3)
         (5 5)
         (6 8)
         (7 13)
         (8 21)
         (9 34))
     (+ (fib (- x 2))
        (fib (- x 1))))))
```

Requirement for Alternative Definition

For ACL2 to accept an alternative (:exec) definition, it must prove, that the two definitions are equal when the input guard is satisfied.

```
(implies
 (natp x)
 (equal
   (if (zp x)
       0
     (if (= x 1)
         1
       (+ (fib (- x 2))
          (fib (- x 1)))))
```

```
(if (< x 10)
    (case x
      (0 0)
      (1 1)
      (2 1)
      (3 2)
      (4 3)
      (5 5)
      (6 8)
      (7 13)
      (8 21)
      (9 34))
  (+ (fib (- x 2))
     (fib (- x 1))))))
```

Execution

Defined functions may be executed.

```
ACL2!>(time$ (fib 50))
; (EV-REC *RETURN-LAST-ARG3* ...) took
; 3.65 seconds realtime, 3.59 seconds runtime
; (16 bytes allocated).
12586269025
```

Wow! Now, this function call takes less than 4 seconds!

By pre-computing the first ten values – the execution speeds up considerably.

Why? Because every execution can stop when `(< X 10)` – and just return the answer from the `CASE` statement.

The `:exec` version of `FIB` includes *memoized* results for when `(< X 10)`.

Memoized Execution

We can tell ACL2 to memoize `FIB` function calls; for instance, when `(< X 40)`.

Defined functions may be executed.

```
ACL2 !>(memoize 'fib :condition '(< x 40))
  [... 65 or so lines elided ...]
```

Now, we can again evaluate our previous `(FIB 50)` call.

```
ACL2 !>(time$ (fib 50))
; (EV-REC *RETURN-LAST-ARG3* ...) took
; 0.00 seconds realtime, 0.00 seconds runtime
; (16 bytes allocated).
12586269025
```

Everything seems great! But, is it?

Memoized Execution, continued

Now, we want to calculate `(FIB 80)`, with the first 40 values memoized.

```
ACL2 !>(time$ (fib 80))
  ; (EV-REC *RETURN-LAST-ARG3* ...) took
  ; 16.90 seconds realtime, 16.82 seconds runtime
  ; (16 bytes allocated).
  23416728348467685
```

Again, we see that it takes a long time. Either, we must pre-compute all the values – and store them for future use – or we need a better approach.

What about using a new algorithm? One that recognizes the relationship between the results?

Can we idenify a recurrence relation in the sequence of results from our `FIB` function?

A New Fibonacci Function

By looking at the first ten values, we see that each entry is the sum of the preceding two entries. The first two values are given as `0` and `1`.

Can we encode this relationship in a new function?

```
(defun f1 (fx-1 fx n-more)
  (declare (xargs :guard (and (natp fx-1)
                              (natp fx)
                              (natp n-more))))
  (if (zp n-more)
      fx
    (f1 fx (+ fx-1 fx) (1- n-more))))
```

Function `F1` has two *registers* (`fx-1` and `fx`) and a third argument (`n-more`) that says how many times to iterate this function.

The Completed New Fibonacci Function

We create a *wrapper* function with the first two values already computed.

```
(defun fib2 (x)
  (declare (xargs :guard (natp x)))
  (if (zp x)
      x
    (f1 0 1 (1- x))))
```

We run some tests to make sure that `FIB` and `FIB2` agree on some values.

```
ACL2 !>(equal (fib 10) (fib2 10))
T
ACL2 !>(equal (fib 30) (fib2 30))
T
```

Looks good! So, can we compute `(FIB2 100)` ? Yes, in an instant!

FIB2 is Equal to FIB

So, we would like to prove this conjecture:

```
(defthm fib2-is-fib
  (implies (natp x)
           (equal (fib2 x)
                  (fib x))))
```

This observation relates the logical definitions of `FIB` and `FIB2`.

- The theorem prover uses the logical definitions to compare these functions.
- For fast evaluation, we use the `FIB2` definition.

So, now how fast is `FIB`, or more to the point, how fast is `FIB2`?

Executing FIB2

Now, we can compute (FIB 1000) easily by computing (FIB2 1000).

```
ACL2 !>(time$ (integer-length (fib2 1000)))
; (EV-REC *RETURN-LAST-ARG3* ...) took
; 0.00 seconds realtime, 0.00 seconds runtime
; (61,200 bytes allocated).
694
```

If we want the complete answer, we can get it (which we split across 5 lines):

```
ACL2 >(FIB2 1000)
434665576869374564356885276750406258025646605173710...
780402481729089536555417949051890403879840079255160...
929592259308032263477520968962323987332247116164290...
964409065331879382989696499285160037044761377951660...
849228875
```

Software Development

Software development can be a very complex undertaking.

We need mechanisms to help us with software development.

We will investigate techniques that will make us better programmers in every computer language we use.

Hopefully, our study will help you.